

# WPROWADZENIE DO MAXIMY

Maxima dostępna jest na stronie [maxima.sourceforge.net](http://maxima.sourceforge.net)

## 1 Obliczenia o dowolnej precyzji

Przypisanie zmiennej  $q$  wartości pierwiastka z dwóch (polecenie zawsze kończy średnik, a wprowadzamy je naciskając **Shift+Enter**)

```
(%i1) q: 2^(1/2);
```

```
(%o1)  $\sqrt{2}$ 
```

Wyznaczenie przybliżonej wartości liczbowej

```
(%i2) q, numer;
```

```
(%o2) 1.414213562373095
```

To samo, ale z dokładnością 80 cyfr

```
(%i3) bfloat(q), fpprec: 80;
```

```
(%o3) 1.414213562373095048801688724209698078569671875376948073176679737990732478462107b0
```

Stała  $\pi$  to w Maximie `%pi`

```
(%i4) bfloat(sin(%pi / 9)), fpprec: 60;
```

```
(%o4) 3.42020143325668733044099614682259580763083367514160628465048b - 1
```

Wartości silni wyznacza się szczególnie prosto

```
(%i5) 60!;
```

```
(%o5) 8320987112741390144276341183223364380754172606361245952449277696409600000000000000
```

## 2 System pomocy

Proszę wypróbować następujące opcje:

```
(%i6) apropos("exp");
```

Poniżej proszę zwrócić uwagę na odstęp po znaku zapytania:

```
(%i7) ? exp;
```

```
(%i8) ?? exp;
```

Poza tym, pomoc uzyskuje się najeżdżając kursorem na nazwę funkcji i naciskając klawisz **F1**.

### 3 Obliczenia na macierzach

Macierz wprowadzamy wierszami:

```
(%i9) mac: matrix([a, b, c], [b, a, c], [c, b, a]);
```

$$(\%o9) \begin{pmatrix} a & b & c \\ b & a & c \\ c & b & a \end{pmatrix}$$

Macierz odwrotna

```
(%i10) invert(mac);
```

$$(\%o10) \begin{pmatrix} \frac{a^2-bc}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} & \frac{bc-ab}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} & \frac{bc-ac}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} \\ \frac{c^2-ab}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} & \frac{a^2-c^2}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} & \frac{bc-ac}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} \\ \frac{b^2-ac}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} & \frac{bc-ab}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} & \frac{a^2-b^2}{b(c^2-ab)+a(a^2-bc)+c(b^2-ac)} \end{pmatrix}$$

Rozłóżmy na czynniki elementy powyższej macierzy (znak % oznacza rezultat ostatniej operacji)

```
(%i11) factor(%);
```

$$(\%o11) \begin{pmatrix} -\frac{bc-a^2}{(b-a)(c-a)(c+b+a)} & \frac{b}{(b-a)(c+b+a)} & \frac{c}{(c-a)(c+b+a)} \\ \frac{c^2-ab}{(b-a)(c-a)(c+b+a)} & -\frac{c+a}{(b-a)(c+b+a)} & \frac{c}{(c-a)(c+b+a)} \\ -\frac{ac-b^2}{(b-a)(c-a)(c+b+a)} & \frac{b}{(b-a)(c+b+a)} & -\frac{b+a}{(c-a)(c+b+a)} \end{pmatrix}$$

Utwórzmy szablon  $(i, j)$ -tego elementu macierzy VanderMonde'a:

```
(%i12) h[i, j] := x[i]^(j - 1);
```

$$(\%o12) h_{i,j} := x_i^{j-1}$$

A teraz samą macierz VanderMonde'a

```
(%i13) VanderMonde: genmatrix(h, 5, 5);
```

$$(\%o13) \begin{pmatrix} 1 & x_1 & x_1^2 & x_1^3 & x_1^4 \\ 1 & x_2 & x_2^2 & x_2^3 & x_2^4 \\ 1 & x_3 & x_3^2 & x_3^3 & x_3^4 \\ 1 & x_4 & x_4^2 & x_4^3 & x_4^4 \\ 1 & x_5 & x_5^2 & x_5^3 & x_5^4 \end{pmatrix}$$

Obliczmy jej wyznacznik:

```
(%i14) detVanderMonde: determinant(VanderMonde);
```

```
(%o14)

$$-x_1(-x_2^2(-x_3^3(x_5^4-x_4^4)+x_4^3x_5^4+x_3^4(x_5^3-x_4^3)-x_4^4x_5^3)+x_2^3(-x_3^2(x_5^4-x_4^4)+x_4^2x_5^4+x_3^4(x_5^2-x_4^2)-x_4^4x_5^2))+..$$

```

A teraz rozłóżmy to długie wyrażenie na czynniki

```
(%i15) factor(detVanderMonde);
```

```
(%o15)

$$(x_2-x_1)(x_3-x_1)(x_3-x_2)(x_4-x_1)(x_4-x_2)(x_4-x_3)(x_5-x_1)(x_5-x_2)(x_5-x_3)(x_5-x_4)$$

```

Utwórzmy szablon macierzy Hilberta

```
(%i16) g[i, j] := 1 / (1 + i + j);
```

```
(%o16)

$$g_{i,j} := \frac{1}{1+i+j}$$

```

Tym razem utwórzmy funkcję, która dla danego  $n$  zwraca macierz Hilberta o tym rozmiarze (do definicji funkcji stosuje się symbol :=)

```
(%i17) Hilbert(n) := genmatrix(g, n, n);
```

```
(%o17)

$$Hilbert(n) := genmatrix(g, n, n)$$

```

Teraz parę wywołań tej funkcji

```
(%i18) invHilbert: invert(Hilbert(4));
```

```
(%o18)

$$\begin{pmatrix} 1200 & -6300 & 10080 & -5040 \\ -6300 & 35280 & -58800 & 30240 \\ 10080 & -58800 & 100800 & -52920 \\ -5040 & 30240 & -52920 & 28224 \end{pmatrix}$$

```

```
(%i19) determinant(Hilbert(4));
```

```
(%o19)

$$\frac{1}{10668672000}$$

```

```
(%i20) %, numer;
```

```
(%o20)

$$9.3732378312877175 \cdot 10^{-11}$$

```

```
(%i21) Hilbert(9);
```

```
(%o21) 
$$\begin{pmatrix} \frac{1}{3} & \frac{1}{4} & \frac{1}{5} & \frac{1}{6} & \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} \\ \frac{1}{4} & \frac{1}{5} & \frac{1}{6} & \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} \\ \frac{1}{5} & \frac{1}{6} & \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} \\ \frac{1}{6} & \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} \\ \frac{1}{7} & \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} \\ \frac{1}{8} & \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} \\ \frac{1}{9} & \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} \\ \frac{1}{10} & \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} \\ \frac{1}{11} & \frac{1}{12} & \frac{1}{13} & \frac{1}{14} & \frac{1}{15} & \frac{1}{16} & \frac{1}{17} & \frac{1}{18} & \frac{1}{19} \end{pmatrix}$$

```

```
(%i22) determinant(%);
```

```
(%o22) 
$$\frac{1}{462068939479146913162956288390362787269836800000000}$$

```

```
(%i23) %, numer;
```


```
(%o23) 
$$2.164179226431492 \cdot 10^{-51}$$

```

## 4 Wykresy funkcji

Wykresy funkcji dwóch zmiennych


```
(%i24) wxplot3d(sin(cos(x * log(y^2 + 1/2))), [x, 0, 2 * %pi], [y, -1, 10])$
```



```
(%t24)
```


Wykresy funkcji jednej zmiennej

```
(%i25) wxplot2d(sin(1 / x) / (x + 1 / x), [x, 10^(-10), 1])$
```



```
(%t25)
```


```
(%i26) wxplot2d(sin(1 / x) / (x + 1 / x), [x, 10^(-10), 0.1])$
```



```
(%t26)
```

Wykresy poziomicowe


```
(%i27) wxcontour_plot(x^3 - y^2 * sin(y), [x, -2, 2], [y, -%pi, %pi]);
```



```
(%t27)
```

```
(%o27)
```

```
(%i28) wxcontour_plot(x^3 - y^2 * sin(y), [x, -2, 2], [y, -%pi, %pi],  
[gnuplot_preamble, "set cntrparam levels 30"]);
```



```
(%t28)
```

```
(%o28)
```

## 5 Faktoryzacja liczb całkowitych i wielomianów

```
(%i29) fermat: 2^2^6 + 1;
```

```
(%o29) 18446744073709551617
```

Rozkład na liczby pierwsze z ich wielokrotnościami

```
(%i30) ifactors(fermat);
```

```
(%o30) [[274177, 1], [67280421310721, 1]]
```

```
(%i31) calkowita: 111541611655546500000005001;
```

```
(%o31) 111541611655546500000005001
```

```
(%i32) ifactors(calkowita);
```

```
(%o32) [[3, 3], [97, 1], [688957, 1], [61817195720387047, 1]]
```

Do wyznaczenia iloczynów służy funkcja `product`

```
(%i33) wiel: product((x + i!), i, 1, 20);
```

```
(%o33) (x + 1) (x + 2) (x + 6) (x + 24) (x + 120) (x + 720) (x + 5040) (x + 40320) (x + 362880) ...
```

Wymnożenia członów dokonuje `expand`

```
(%i34) wiel_rozl: expand(wiel);
```

```
(%o34) x^20 + 2561327494111820313 x^19 + 313273851453819662344483252264783076 x^18 + ...
```

Ponownej faktoryzacji dokona `factor`

```
(%i35) factor(wiel_rozl);
```

```
(%o35) (x + 1) (x + 2) (x + 6) (x + 24) (x + 120) (x + 720) (x + 5040) (x + 40320) ...
```

```
(%i36) factor(x^3 + 125); factor(x^3 + 216);
```

```
(%o36) (x + 5) (x^2 - 5 x + 25)
```

```
(%o37) (x + 6) (x^2 - 6 x + 36)
```

```
(%i38) factor(x^4 + 2500); factor(x^4 + 5184);
```

```
(%o38) (x^2 - 10 x + 50) (x^2 + 10 x + 50)
```

```
(%o39) (x^2 - 12 x + 72) (x^2 + 12 x + 72)
```

```
(%i40) factor(x^5 + 1); factor(x^5 + 32);
```

```
(%o40) (x + 1) (x^4 - x^3 + x^2 - x + 1)
```

```
(%o41) (x + 2) (x^4 - 2 x^3 + 4 x^2 - 8 x + 16)
```

## 6 Upraszczanie wyrażeń trygonometrycznych

```
(%i42) atryg: cos(3 * x)^2 * sin(2 * x)^3;
```

$$(\%o42) \quad \sin(2x)^3 \cos(3x)^2$$

Pozbycie się wielokrotności kątów

```
(%i43) trigexpand(atrig);
```

$$(\%o43) \quad 8 \cos(x)^3 \sin(x)^3 \left( \cos(x)^3 - 3 \cos(x) \sin(x)^2 \right)^2$$

I znowu powrót do kątów wielokrotnych

```
(%i44) trigreduce(%);
```

$$(\%o44) \quad \frac{-\sin(12x) - 2\sin(6x)}{16} + \frac{3\sin(8x) - 3\sin(4x) + 6\sin(2x)}{16}$$

W ten sposób otrzymuje się tzw. postać kanoniczną:

```
(%i45) trigrat(atrig);
```

$$(\%o45) \quad -\frac{\sin(12x) - 3\sin(8x) + 2\sin(6x) + 3\sin(4x) - 6\sin(2x)}{16}$$

```
(%i46) btryg: cos(2 * x)^2 + 3 + sin(2 * x)^2;
```

$$(\%o46) \quad \sin(2x)^2 + \cos(2x)^2 + 3$$

Uproszczenie z zastosowaniem „jedynki trygonometrycznej”

```
(%i47) trigsimp(btryg);
```

$$(\%o47) \quad 4$$

## 7 Całkowanie i różniczkowanie

```
(%i48) funk: x^2 * sin(x / 5) + x - 1;
```

$$(\%o48) \quad x^2 \sin\left(\frac{x}{5}\right) + x - 1$$

Całka nieoznaczona  $\int \dots dx$ :

```
(%i49) integrate(funk, x);
```

```
(%o49)
```

$$125 \left( \frac{2x \sin\left(\frac{x}{5}\right)}{5} + \left(2 - \frac{x^2}{25}\right) \cos\left(\frac{x}{5}\right) \right) + \frac{x^2}{2} - x$$

Rezultat sprawdzamy różniczkując go

```
(%i50) diff(%, x);
```

```
(%o50)
```

$$125 \left( \frac{2 \sin\left(\frac{x}{5}\right)}{5} - \frac{\left(2 - \frac{x^2}{25}\right) \sin\left(\frac{x}{5}\right)}{5} \right) + x - 1$$

```
(%i51) trigsimp(%)
```

```
(%o51)
```

$$x^2 \sin\left(\frac{x}{5}\right) + x - 1$$

Całka oznaczona  $\int_1^\infty \frac{1}{x^2} dx$

```
(%i52) integrate(1 / x^2, x, 1, inf);
```

```
(%o52)
```

$$1$$

Czasem całka może po prostu nie istnieć

```
(%i53) integrate(1 / x, 0, inf);
```

*Attempt to integrate wrt a number: 0 – an error. To debug this try: debugmode(true);*

```
(%i54) g: x^3 * %e^(x) / sin(x)^4;
```

```
(%o54)
```

$$\frac{x^3 e^x}{\sin(x)^4}$$

Rozwinięcie w szereg Taylora

```
(%i55) taylor(g, x, 0, 3);
```

```
(%o55)
```

$$\frac{1}{x} + 1 + \frac{7x}{6} + \frac{5x^2}{6} + \frac{223x^3}{360} + \dots$$

Granica funkcji  $\lim_{x \rightarrow 0} \frac{\sin(x)}{x} dx$

```
(%i56) limit(sin(x) / x, x, 0);
```

```
(%o56)
```

$$1$$



## 8 Rozwiązywanie równań algebraicznych

```
(%i57) solve([3 * x - 5 * y = 2, 8 * x + 2 * y = 3]);
```

```
(%o57) 
$$\left[ \left[ y = -\frac{7}{46}, x = \frac{19}{46} \right] \right]$$

```

```
(%i58) solve(x^2 - 5 * x - 5 = 0);
```

```
(%o58) 
$$\left[ x = -\frac{3\sqrt{5}-5}{2}, x = \frac{3\sqrt{5}+5}{2} \right]$$

```

Wyznaczmy przybliżenie numeryczne pierwszego pierwiastka

```
(%i59) part(%, 1, 2), numer;
```

```
(%o59) 
$$-0.85410196624968$$

```

Nie każde równanie można rozwiązać analitycznie...


```
(%i60) solve(exp(-x) - sin(x) = 0);
```

```
(%o60) 
$$[\sin(x) = e^{-x}]$$

```

... chociaż pierwiastek istnieje

```
(%i61) wxplot2d(exp(-x) - sin(x), [x, -1, 1])$
```



```
(%t61)
```

Wtedy można wyznaczyć pierwiastek w sposób przybliżony

```
(%i62) find_root(exp(-x) - sin(x), x, 0, 1);
```

```
(%o62) 
$$0.58853274398186$$

```